

# Dubbels

Dubbelgetallen (double numbers)

## Vorm

Een dubbel beslaat 2 cellen.

## Inhoud

Een dubbel kan zijn

- één getal met dubbele precisie
- een getallenpaar
- of ...?

## Manoeuvreren met dubbels

Alleen de vorm is van belang, de '2' staat voor '2 cellen'.

```
: 2DROP ( x y -- ) drop drop ;
: 2DUP  ( x y -- x y x y )
: 2SWAP ( x y a b -- a b x y )
: 2OVER ( x y a b -- x y a b x y )
: 2NIP  ( x y a b -- a b )
: 2TUCK ( x y a b -- a b x y a b )
2@ 2!
2CONSTANT 2VARIABLE
2LITERAL
2>R 2R> 2R@
```

Let op de volgorde van de cellen:

```
: SWAP ( x y -- y x ) 2>r r> r> ;
: SWAP ( x y -- y x ) >r >r 2r> ;
```

```
ROT ( a m n -- m n a ) \ MSWAP
-ROT ( m n a -- a m n ) \ -MSWAP
```

## Dubbele precisie

Eén getal beslaat 2 cellen. De 'D' geeft aan dat het om dubbele precisie gaat. In het geheugen staat het hoge deel (hi) altijd voorop.

```
: S>D ( x -- x xhi ) dup 0< ;
: D>S ( xlo xhi -- xlo ) drop ;
```

## Rekenen

De M staat voor Mixed, dwz. een operatie met zowel 1-cellige als 2-cellige getallen.

```
D+ D- DNEGATE D2* D2/ DMAX DMIN
*/ UM/MOD UM* M*/
```

## Vergelijken

```
D= D0= D< D0< DU<
```

Waarom geen UD< ?

## Uitvoer

```
D. D.R
```

Waarom geen DU. (of UD.) ?

## Invoer

```
12345.
```

## Breuken (getallenparen)

Quasi floating point met \*/ (en een juiste schaling)

```
decimal 100 3 4 */ . ( levert 75 )
```

### PI

```
22 / 7                3.14~
355 / 113             3.141 592 ~
104348 / 33215       3.141 592 653 ~
4272943 / 1360120    3.141 592 653 589 ~
21053343141 / 6701487259 3.141 592 653 589 793 238 462 ~
```

```
: PI* ( x -- x*pi ) 355 113 */ ;
: C-OPPERV ( straal -- opperv ) dup * pi* ; \  $\pi r^2$ 
: C-OMTREK ( straal -- omtrek ) 2* pi* ; \  $2\pi r$ 
```

### Test

```
: %A ( x percentage -- y ) >r 100 / r> * ;
: %B ( x percentage -- y ) * 100 / ;
: %C ( x percentage -- y ) 100 */ ;
```

```
17          constant KG \ klein getal
777         constant MG \ getal
-1 1 rshift constant GG \ groot getal
```

```
kg          .
kg 50 %a    .
kg 50 %b    .
kg 50 %c    .
```

```
mg          .
mg 50 %a    .
mg 50 %b    .
mg 50 %c    .
```

```
gg          .
gg 50 %a    .
gg 50 %b    .
gg 50 %c    .
```

### Resultaten

#### 16bits

```
:)kg          . 17 OK.0
:)kg 50 %a    . 0 OK.0
:)kg 50 %b    . 8 OK.0
:)kg 50 %c    . 8 OK.0
```

```
:)mg          . 777 OK.0
:)mg 50 %a    . 350 OK.0
:)mg 50 %b    . -267 OK.0
:)mg 50 %c    . 388 OK.0
```

```
:)gg          . 32767 OK.0
:)gg 50 %a    . 16350 OK.0
:)gg 50 %b    . -1 OK.0
:)gg 50 %c    . 16383 OK.0
```

#### 32bits

```
:)kg          . 17 OK.0
:)kg 50 %a    . 0 OK.0
:)kg 50 %b    . 8 OK.0
:)kg 50 %c    . 8 OK.0
```

```
:)mg          . 777 OK.0
:)mg 50 %a    . 350 OK.0
:)mg 50 %b    . 388 OK.0
:)mg 50 %c    . 388 OK.0
```

```
:)gg          . 2147483647 OK.0
:)gg 50 %a    . 1073741800 OK.0
:)gg 50 %b    . -1 OK.0
:)gg 50 %c    . 1073741823 OK.0
```

## Celsius en Fahrenheit

```
(C * 1,8 + 32) = F
(F - 32) / 1,8 = C
```

```
decimal
: C>F ( C -- F ) 9 5 */ 32 + ;
: F>C ( F -- C ) 32 - 5 9 */ ;
```

Voorbeelden

```
:)0 f>c . -18 OK.0
:)0 c>f . 32 OK.0
:)100 f>c . 37 OK.0
:)37 c>f . 98 OK.0
```

In tienden van graden (schaling)

```
: F>>C ( F*10 -- C*10 ) 320 - 5 9 */ ;
: C>>F ( C*10 -- F*10 ) 9 5 */ 320 + ;
```

Voorbeelden

```
:)0 f>>c 1 .f -17.8 OK.0
:)0 c>>f 1 .f 32.0 OK.0
:)1000 f>>c 1 .f 37.7 OK.0
:)370 c>>f 1 .f 98.6 OK.0
```

## Getallen afdrucken met vaste komma

```
: (FIX) ( du i -- du' ) 20 over u< ?abort
  <# for # next ch . hold #s ;
: DU.F ( du i -- a n ) (fix) #> ;
: D.F ( dx i -- a n ) >r tuck dabs r> (fix) rot sign #> ;

: D.F ( dx i -- ) d.f) type space ;
: DU.F ( du i -- ) du.f) type space ;
: .F ( x i -- ) over 0< swap d.f ;
: U.F ( u i -- ) 0 swap du.f ;
```

Voorbeelden

```
@)-1 4 .f -0.0001 OK.0
@)-1 5 u.f FFF.FFFFF OK.0
@)-1. 6 d.f -0.000001 OK.0
@)-1. 7 du.f FFFFFFFF.FFFFFFFF OK.0
```

## Complexe getallen x+yi

```
: C+ ( x y x1 y1 -- x2 y2 ) rot + >r + r> ;
: CNEGATE ( x y -- x1 y1 ) negate >r negate r> ;
: C- ( x y x1 y1 -- x2 y2 ) cnegate c+ ;
: C* ( x y x1 y2 -- x2 y2 )
  2over 2over
  >r * swap r> * + >r \ imaginaire deel
  >r rot * swap r> * - \ reele deel
  r> ;
: .C ( x y -- ) (.)
  swap 0 .r
  s>d 0= if ch + emit then 0 .r
  ch i emit ;
\ vb
\ 10 5 .c ( levert 10+5i )
```

\ <><>